# Parallella Firmware Files
# Maintainer: Kevin Mihelich <kevin@archlinuxarm.org>
# Modified: Scott Tincman <sctincman@gmail.com>

buildarch=4

pkgbase=parallella-firmware
pkgname=('parallella-firmware-headless-7010' 'parallella-firmware-headless-7020')
pkgver=2015.1
pkgrel=1
pkgdesc="Bitstream files for Parallella"
provides=("parallella-firmware")
conflicts=("parallella-firmware")
arch=('armv7h')
url="https://github.com/parallella/parallella-hw"
license=('GPL')
_commit=beb4ca09e9616cc76364735f19e8502c438cd61c
source=("https://github.com/parallella/parallella-hw/raw/${_commit}/fpga/bitstreams/parallella_e16_headless_gpiose_7010.bit.bin"
	"https://github.com/parallella/parallella-hw/raw/${_commit}/fpga/bitstreams/parallella_e16_headless_gpiose_7020.bit.bin")
sha256sums=('507d6241908bcea2150eaf45b2c723b3827d89b7822a19c2bc37d219b24f3c3a'
            '5e8f72ce0dc809fdf039e00f260a4a37db6a1278d41d502d1432725ab6313bc5')

package_parallella-firmware-headless-7010() {
    pkgdesc="Bitstream files needed for Parallella (Headless, Zynq 7010)"
    install -m 0755 -d "${pkgdir}"/boot

    install -m 0755 parallella_e16_headless_gpiose_7010.bit.bin "${pkgdir}"/boot/parallella.bit.bin
}

package_parallella-firmware-headless-7020() {
    pkgdesc="Bitstream files needed for Parallella (Headless, Zynq 7020)"
    install -m 0755 -d "${pkgdir}"/boot

    install -m 0755 parallella_e16_headless_gpiose_7020.bit.bin "${pkgdir}"/boot/parallella.bit.bin
}
